Title

stata.com

graph twoway pci — Twoway paired-coordinate plot with immediate arguments

Description

`pci` is an immediate version of `twoway pcspike`; see **[U] 19 Immediate commands** and [G-2] **graph twoway pcspike**. `pci` is intended for programmer use but can be useful interactively.

Quick start

A paired-coordinate plot from $(21, 40)$ to $(22, 36)$

```
twoway pci 40 21 36 22
```

Draw two line segments such that they appear to form a single line

```
twoway pci 40 21 36 22 || pci 36 22 39 24
```

Specify both line segments are black

```
twoway pci 40 21 36 22 || pci 36 22 39 24, lcolor(black black)
```

Add the label "My label" to the line at 6 o'clock by recasting as `pccapsym`

```
twoway pci 36 22 39 24 (6) "My label", recast(pccapsym) msymbol(i)
```

Menu

Graphics > Twoway graph (scatter, line, etc.)

Syntax

```
twoway pci immediate_values [ , options ]
```

where *immediate_values* is one or more of

$\#_{y1}$ $\#_{x1}$ $\#_{y2}$ $\#_{x2}$ [(#clockposstyle)] ["*text for label*"]

See [G-4] ***clockposstyle*** for a description of #clockposstyle.

Options

options are as defined in [G-2] **graph twoway pcspike**, with the following modifications:

If "*text for label*" is specified among any of the immediate arguments, option `mlabel()` is assumed.

If (#clockposstyle) is specified among any of the immediate arguments, option `mlabvposition()` is assumed.

Also see the *marker_options* defined in [G-2] **graph twoway pccapsym** if the `recast()` option is used to change the spikes into a paired-coordinate plot that plots markers.

Remarks and examples

stata.com

Immediate commands are commands that obtain data from numbers typed as arguments.

`twoway pci` does not modify the data in memory.

`pci` is intended for programmer use but can be used interactively. We can combine a `pci` plot with other `twoway` plots to produce a quick diagram.

```
. twoway function  y = -x^2, range(-2 2)          ||
         pci 0 1 0 -1                             ||
         pcarrowi 1.2 .5 0 0
```



We can improve the annotation with

```
. twoway function  y = -x^2, range(-2 2)                        ||
         pci 0 1 0 -1 "Tangent", recast(pccapsym) msymbol(i) ||
         pcarrowi 1.2 .5 0.05 0 "Maximum at x=0",
         legend(off) title("Characteristics of y = -x{superscript:2}")
```



A slightly more whimsical example is

```
. twoway pci 2 0 2 6  4 0 4 6  0 2 6 2  0 4 6 4 ||
         scatteri 5 1  3 3, msize(ehuge) ms(X)  ||
         scatteri 5 5  1 5, msize(ehuge) ms(Oh) legend(off)
```



❑ Technical note

Programmers: Note carefully `twoway`'s *advanced_option* `recast()`; see [G-3] ***advanced_options***. It can be used to good effect, such as using `pci` to add marker labels.

❑

Also see

[G-2] **graph twoway** — Twoway graphs

[G-2] **graph twoway pcarrow** — Paired-coordinate plot with arrows

[G-2] **graph twoway scatteri** — Scatter with immediate arguments

[U] **19 Immediate commands**